# APPENDIX: PEAK-LOCUS THEOREM proved for ALL Lognormal Stochastic Processes L(t) starting at t = ts with probability one.

Author: Claudio Maccone (e-mail: clmaccon@libero.it),
Key equations discovered on December 21, 22, 23, 24, 2013. This code version completed on January 20, 2014.

```
(%i1) kill(all);
```
(%o0) *done*

Defining the lognormal pdf of the lognormal stochastic process L(t) STARTING AT t = ts (with ts = t_start).

```
(%i1) assume(t>ts, te>ts, sL>0);
```
(%o1) $[t>ts, te>ts, sL>0]$

```
(%i2) depends(M,t);
```
(%o2) $[M(t)]$

```
(%i3) Lpdf(n,sL,ts,t):=(%e^(-(log(n)-M(t))^2/(2*sL^2*(t-ts))))/(sqrt(2*%pi)*sL*n*sqrt(t-ts));
```
(%o3) $\mathrm{Lpdf}(n,sL,ts,t):=\frac{\%e^{-\frac{(\log(n)-M(t))^2}{2\,sL^2\,(t-ts)}}}{\sqrt{2\,\pi}\,sL\,n\,\sqrt{t-ts}}$

Normalization condition.

```
(%i4) normalization_condition:'integrate(Lpdf(n,sL,ts,t),n,0,inf)=radcan(integrate(Lpdf(n,sL,ts,t),n,0,inf));
```
(%o4) $\frac{\int_0^{\infty}\frac{\%e^{-\frac{(\log(n)-M(t))^2}{2\,(t-ts)\,sL^2}}}{n}\,dn}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{t-ts}\,sL}=1$

Mean value.

```
(%i5) def_mean_value_integral:'integrate(n*Lpdf(n,sL,ts,t),n,0,inf)=radcan(integrate(n*Lpdf(n,sL,ts,t),n,0,inf));
```
(%o5) $\frac{\int_0^{\infty}\%e^{-\frac{(\log(n)-M(t))^2}{2\,(t-ts)\,sL^2}}\,dn}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{t-ts}\,sL}=\%e^{-\frac{(ts-t)\,sL^2-2\,M(t)}{2}}$

(%i6) easy_mean_value_integral:lhs(def_mean_value_integral)=factor(rhs(def_mean_value_integral));

(%o6) $$\frac{\int_{0}^{\infty} \%e^{-\frac{(\log(n)-M(t))^2}{2\,(t-ts)\,sL^2}}\,dn}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{t-ts}\,sL}=\%e^{-\frac{ts\,sL^2}{2}+\frac{t\,sL^2}{2}+M(t)}$$

(%i7) def_mean_value:mean_value=rhs(easy_mean_value_integral);

(%o7) $$mean\_value=\%e^{-\frac{ts\,sL^2}{2}+\frac{t\,sL^2}{2}+M(t)}$$

Introducing the SCALE FACTOR (denoted K) on the VERTICAL SCALE of the stochastic process L(t). Defining the TEMPORARY MEAN VALUE FUNCTION OF THE TIME m(t).

(%i8) temporary_m(t):=K*%e^(-(ts*sL^2)/2+(t*sL^2)/2+M(t));

(%o8) $$\mathrm{temporary\_m}(t):=K\,\%e^{\frac{-ts\,sL^2}{2}+\frac{t\,sL^2}{2}+M(t)}$$

Assigning the starting value Ns (with probability 1) of the process L(t) at the starting time t=ts.

(%i9) def_Ns:temporary_m(ts)=Ns;

(%o9) $$\%e^{M(ts)}\,K=Ns$$

(%i10) expression_of_K:first(solve(def_Ns,K));

(%o10) $$K=Ns\,\%e^{-M(ts)}$$

Defining the MEAN VALUE FUNCTION OF THE TIME m(t).

(%i11) temporary_m(t),expression_of_K;

(%o11) $$Ns\,\%e^{-\frac{ts\,sL^2}{2}+\frac{t\,sL^2}{2}-M(ts)+M(t)}$$

(%i12) m(t):=Ns*%e^(M(t)-M(ts))*%e^((sL^2/2)*(t-ts));

(%o12) $$\mathrm{m}(t):=Ns\,\%e^{M(t)-M(ts)}\,\%e^{\frac{sL^2}{2}(t-ts)}$$

Checking the special case t = ts, namely the INITIAL CONDITION with probability one.

(%i13) m(ts);

(%o13) $Ns$

Defining the k-th MOMENT of the general Lognormal Stochastic Process L(t).

(%i14) declare(k, integer);

(%o14) *done*

(%i15) k_moment:'integrate(n^k*Lpdf(n,sL,ts,t),n,0,inf)=expand(radcan(integrate(n^k*Lpdf(n,sL,ts,t),n,0,inf)));

*Is* $k$ *positive, negative, or zero?* p;

*Is* $k-1$ *positive, negative, or zero?* p;

(%o15) $$\frac{\int_0^{\infty} n^{k-1}\,\%e^{-\frac{(\log(n)-M(t))^2}{2\,(t-ts)\,sL^2}}\,dn}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{t-ts}\,sL}=\%e^{-\frac{k^2\,ts\,sL^2}{2}+\frac{k^2\,t\,sL^2}{2}+k\,M(t)}$$

(%i16) def_mean_value_of_the_square:subst(2, k, rhs(k_moment));

(%o16) $$\%e^{-2\,ts\,sL^2+2\,t\,sL^2+2\,M(t)}$$

(%i17) def_variance:variance=factor(expand(def_mean_value_of_the_square-(rhs(def_mean_value))^2));

(%o17) $$variance=-(\%e^{ts\,sL^2}-\%e^{t\,sL^2})\,\%e^{-2\,ts\,sL^2+t\,sL^2+2\,M(t)}$$

(%i18) def_st_dev:standard_devation=sqrt(rhs(def_variance));

(%o18) $$standard\_devation=\sqrt{\%e^{t\,sL^2}-\%e^{ts\,sL^2}}\;\%e^{\frac{-2\,ts\,sL^2+t\,sL^2+2\,M(t)}{2}}$$

Finding the upper and lower standard deviation curves.

(%i19) def_upper_std_curve:upper_std_curve(p)=(expand(rhs(def_mean_value)+rhs(def_st_dev)));

(%o19) $$\text{upper\_std\_curve}(p)=\%e^{-\frac{ts\,sL^2}{2}+\frac{t\,sL^2}{2}+M(t)}+\sqrt{\%e^{t\,sL^2}-\%e^{ts\,sL^2}}\;\%e^{-ts\,sL^2+\frac{t\,sL^2}{2}+M(t)}$$

(%i20) def_lower_std_curve:lower_std_curve(p)=(expand(rhs(def_mean_value)-rhs(def_st_dev)));

(%o20) $$\text{lower\_std\_curve}(p)=\%e^{-\frac{ts\,sL^2}{2}+\frac{t\,sL^2}{2}+M(t)}-\sqrt{\%e^{t\,sL^2}-\%e^{ts\,sL^2}}\;\%e^{-ts\,sL^2+\frac{t\,sL^2}{2}+M(t)}$$

Defining the standard deviation function of t: Delta(t).

(%i21) Delta(t):=sqrt(%e^(t*sL^2)-%e^(ts*sL^2))*%e^((-2*ts*sL^2+t*sL^2+2*M(t))/2);

(%o21) $$\Delta(t):=\sqrt{\%e^{t\,sL^2}-\%e^{ts\,sL^2}}\;\%e^{\frac{(-2)\,ts\,sL^2+t\,sL^2+2\,M(t)}{2}}$$

Finding sL when Ns, Ne and Delta(te) are known.

(%i22) def_of_standard_deviation_at_end:DeltaNe=Delta(te);

(%o22) $DeltaNe=\sqrt{\%e^{te\,sL^2}-\%e^{ts\,sL^2}}\,\%e^{\frac{-2\,ts\,sL^2+te\,sL^2+2\,M(te)}{2}}$

Final Boundary condition.

(%i23) final_boundary_condition_eq:Ne=m(te);

(%o23) $Ne=Ns\,\%e^{\frac{(te-ts)\,sL^2}{2}-M(ts)+M(te)}$

Finding sL in terms of the four Boundary conditions: (ts, Ns) and (te, Ne),
PLUS the additional fifth conditon on the final standard deviation: DeltaNe.

(%i24) towards_sL_1:def_of_standard_deviation_at_end/final_boundary_condition_eq;

(%o24) $\frac{DeltaNe}{Ne}=\frac{\sqrt{\%e^{te\,sL^2}-\%e^{ts\,sL^2}}\,\%e^{\frac{-2\,ts\,sL^2+te\,sL^2+2\,M(te)}{2}-\frac{(te-ts)\,sL^2}{2}+M(ts)-M(te)}}{Ns}$

(%i25) towards_sL_2:expand((expand(towards_sL_1*Ns))^2)+%e^(2*M(ts));

(%o25) $\%e^{2\,M(ts)}+\frac{DeltaNe^2\,Ns^2}{Ne^2}=\%e^{-ts\,sL^2+te\,sL^2+2\,M(ts)}$

(%i26) towards_sL_3:log(towards_sL_2);

(%o26) $\log\left(\%e^{2\,M(ts)}+\frac{DeltaNe^2\,Ns^2}{Ne^2}\right)=-ts\,sL^2+te\,sL^2+2\,M(ts)$

(%i27) def_sL:second(solve(towards_sL_3,sL));

(%o27) $sL=\frac{\sqrt{\log\left(\%e^{2\,M(ts)}+\frac{DeltaNe^2\,Ns^2}{Ne^2}\right)-2\,M(ts)}}{\sqrt{te-ts}}$

Defining the RUNNING b-LOGNORMAL.

(%i28) assume(sigma>0);

(%o28) $[\sigma>0]$

(%i29) RbLpdf(t, mu, sigma, b):=(%e^(-(log(t-b)-mu)^2/(2*sigma^2)))/(sqrt(2*%pi)*sigma*(t-b));

(%o29) $\mathrm{RbLpdf}(t,\mu,\sigma,b):=\frac{\%e^{\frac{-(\log(t-b)-\mu)^2}{2\,\sigma^2}}}{\sqrt{2\,\pi}\,\sigma\,(t-b)}$

(%i30) assume(mu>0);

(%o30) $[\mu>0]$

(%i31) 'integrate(RbLpdf(t, mu, sigma, b), t, b, inf)=integrate(RbLpdf(t, mu, sigma, b), t, b, inf);

$$(\%o31)\ \frac{\int_b^{\infty} \frac{\%e^{-\frac{(\log(t-b)-\mu)^2}{2\sigma^2}}}{t-b}\, dt}{\sqrt{2}\sqrt{\pi}\,\sigma}=1$$

(%i32) RbL_PEAK_ABSCISSA:first(solve(diff(RbLpdf(t, mu, sigma, b), t)=0, t));

(%o32) $t=\%e^{\mu-\sigma^2}+b$

(%i33) RbL_p:subst(p,t,RbL_PEAK_ABSCISSA);

(%o33) $p=\%e^{\mu-\sigma^2}+b$

(%i34) RbL_PEAK_ORDINATE:RbLpdf(t, mu, sigma, b),RbL_PEAK_ABSCISSA;

$$(\%o34)\ \frac{\%e^{\frac{\sigma^2}{2}-\mu}}{\sqrt{2}\sqrt{\pi}\,\sigma}$$

Invoking the CONDITION that the RbL_PEAK_ORDINATE must equal
the Exponential Mean Value of the ENVELOPING GBM.

(%i35) assume(p>ts);

(%o35) $[p>ts]$

(%i36) def_RbL_CONDITION:RbL_PEAK_ORDINATE=m(p);

$$(\%o36)\ \frac{\%e^{\frac{\sigma^2}{2}-\mu}}{\sqrt{2}\sqrt{\pi}\,\sigma}=Ns\ \%e^{\frac{(p-ts)\,sL^2}{2}-M(ts)+M(p)}$$

SEPARATING the mu&sigma(p) part from the sigma_only(p) part.

(%i37) Two:[%e^(sigma^2/2-mu)=%e^(p*sL^2/2), 1/(sqrt(2*%pi)*sigma)=Ns*%e^(-(ts*sL^2)/2-M(ts)+M(p))];

$$(\%o37)\ [\%e^{\frac{\sigma^2}{2}-\mu}=\%e^{\frac{p\,sL^2}{2}},\frac{1}{\sqrt{2}\sqrt{\pi}\,\sigma}=Ns\ \%e^{-\frac{ts\,sL^2}{2}-M(ts)+M(p)}]$$

Solving the sigma_only(p) part, namely the second equation, with respect to sigma.

(%i38) final_sigma:first(solve(second(Two), sigma));

(%o38) $\sigma = \frac{\%e^{\frac{ts\ sL^2}{2} + M(ts) - M(p)}}{\sqrt{2}\sqrt{\pi}\ Ns}$

Inserting this into the first equation, we now find the mu(p) function.

(%i39) eq_in_mu_only:first(Two),final_sigma;

(%o39) $\%e^{\frac{\%e^{2\left(\frac{ts\ sL^2}{2} + M(ts) - M(p)\right)}}{4\pi Ns^2} - \mu} = \%e^{\frac{p\ sL^2}{2}}$

(%i40) mu_from_first_eq:distrib(first(solve(log(eq_in_mu_only), mu)));

(%o40) $\mu = \frac{\%e^{ts\ sL^2 + 2M(ts) - 2M(p)}}{4\pi Ns^2} - \frac{p\ sL^2}{2}$

(%i41) final_mu:mu_from_first_eq,sigma_from_second_eq;

(%o41) $\mu = \frac{\%e^{ts\ sL^2 + 2M(ts) - 2M(p)}}{4\pi Ns^2} - \frac{p\ sL^2}{2}$

Summing up: these are mu and sigma of the Running b-Lognormal
for the GENERAL LOGNORMAL STOCHASTIC PROCESS L(t).

(%i42) RbL_mu_and_sigma:[final_mu, final_sigma];

(%o42) $[\mu = \frac{\%e^{ts\ sL^2 + 2M(ts) - 2M(p)}}{4\pi Ns^2} - \frac{p\ sL^2}{2}, \sigma = \frac{\%e^{\frac{ts\ sL^2}{2} + M(ts) - M(p)}}{\sqrt{2}\sqrt{\pi}\ Ns}]$

To these, we must add the value of b(p) obtained from the b-lognormal PEAK ABSCISSA equation:

(%i43) b_from_mu_and_sigma:first(solve(RbL_p,b));

(%o43) $b = p - \%e^{\mu - \sigma^2}$

In CONCLUSION, we have proven the PEAK-LOCUS THEOREM:

(%i44) PEAK_LOCUS_THEOREM:[final_mu,final_sigma,b_from_mu_and_sigma];

(%o44) $[\mu = \frac{\%e^{ts\ sL^2 + 2M(ts) - 2M(p)}}{4\pi Ns^2} - \frac{p\ sL^2}{2}, \sigma = \frac{\%e^{\frac{ts\ sL^2}{2} + M(ts) - M(p)}}{\sqrt{2}\sqrt{\pi}\ Ns}, b = p - \%e^{\mu - \sigma^2}]$

ENTROPY of the Running b-Lognormal in bits.

(%i45) def_H_of_RbL_in_bits:H_of_RbL_in_bits=(log(sqrt(2*%pi)*sigma)+mu+1/2)/log(2);

(%o45) $$H\_of\_RbL\_in\_bits=\frac{\log(\sqrt{2}\,\sqrt{\pi}\,\sigma)+\mu+\frac{1}{2}}{\log(2)}$$

(%i46) Hv1:def_H_of_RbL_in_bits,PEAK_LOCUS_THEOREM;

(%o46) $$H\_of\_RbL\_in\_bits=\frac{\log\left(\frac{\%e^{\frac{ts\,sL^2}{2}+M(ts)-M(p)}}{Ns}\right)+\frac{\%e^{ts\,sL^2+2\,M(ts)-2\,M(p)}}{4\,\pi\,Ns^2}-\frac{p\,sL^2}{2}+\frac{1}{2}}{\log(2)}$$

(%i47) Hv2:expand(radcan(Hv1));

(%o47) $$H\_of\_RbL\_in\_bits=\frac{\%e^{ts\,sL^2+2\,M(ts)-2\,M(p)}}{4\,\pi\,\log(2)\,Ns^2}+\frac{ts\,sL^2}{2\log(2)}-\frac{p\,sL^2}{2\log(2)}+\frac{M(ts)}{\log(2)}-\frac{M(p)}{\log(2)}-\frac{\log(Ns)}{\log(2)}+\frac{1}{2\log(2)}$$

(%i48) Hv3:%e^(ts*sL^2+2*M(ts)-2*M(p))/(4*%pi*log(2)*Ns^2)-(p-ts)*sL^2/(2*log(2))-(M(p)-M(ts))/log(2)-log

(%o48) $$\frac{\%e^{ts\,sL^2+2\,M(ts)-2\,M(p)}}{4\,\pi\,\log(2)\,Ns^2}-\frac{(p-ts)\,sL^2}{2\log(2)}+\frac{M(ts)-M(p)}{\log(2)}-\frac{\log(Ns)}{\log(2)}+\frac{1}{2\log(2)}$$

(%i49) radcan(Hv3-rhs(Hv2));

(%o49) 0

(%i50) H(p):=%e^(ts*sL^2+2*M(ts)-2*M(p))/(4*%pi*log(2)*Ns^2)-((p-ts)*sL^2)/(2*log(2))+(M(ts)-M(p))/log(2

(%o50) $$H(p):=\frac{\%e^{ts\,sL^2+2\,M(ts)+(-2)\,M(p)}}{4\,\pi\,\log(2)\,Ns^2}-\frac{(p-ts)\,sL^2}{2\log(2)}+\frac{M(ts)-M(p)}{\log(2)}+\frac{-\log(Ns)}{\log(2)}+\frac{1}{2\log(2)}$$

(%i51) H(ts);

(%o51) $$\frac{\%e^{ts\,sL^2}}{4\,\pi\,\log(2)\,Ns^2}-\frac{\log(Ns)}{\log(2)}+\frac{1}{2\log(2)}$$

(%i52) def_Evo_Index_in_bits:Evo_Index_in_bits=-H(p)+H(ts);

(%o52) $$Evo\_Index\_in\_bits=-\frac{\%e^{ts\,sL^2+2\,M(ts)-2\,M(p)}}{4\,\pi\,\log(2)\,Ns^2}+\frac{\%e^{ts\,sL^2}}{4\,\pi\,\log(2)\,Ns^2}+\frac{(p-ts)\,sL^2}{2\log(2)}-\frac{M(ts)-M(p)}{\log(2)}$$

(%i53) v2_Evo_Index_in_bits:(%e^(ts*sL^2)*(1-%e^(-2*(M(p)-M(ts)))))/(4*%pi*log(2)*Ns^2)+((p-ts)*sL^2)/(2*

(%o53) $$\frac{(1-\%e^{-2\,(M(p)-M(ts))})\,\%e^{ts\,sL^2}}{4\,\pi\,\log(2)\,Ns^2}+\frac{(p-ts)\,sL^2}{2\log(2)}+\frac{M(p)-M(ts)}{\log(2)}$$

(%i54) radcan(rhs(def_Evo_Index_in_bits)-v2_Evo_Index_in_bits);
(%o54) 0

Thus, the DEFINITION of the function Evo_Index(p) is given by:

(%i55) Evo_Index(p):=((1-%e^(-2*(M(p)-M(ts))))*%e^(ts*sL^2))/(4*%pi*log(2)*Ns^2)+((p-ts)*sL^2)/(2*log(2))

$$(\%o55)\ \mathrm{Evo\_Index}(p):=\frac{\left(1-\%e^{(-2)\,(\mathrm{M}(p)-\mathrm{M}(ts))}\right)\%e^{ts\,sL^2}}{4\,\pi\,\log(2)\,Ns^2}+\frac{(p-ts)\,sL^2}{2\,\log(2)}+\frac{\mathrm{M}(p)-\mathrm{M}(ts)}{\log(2)}$$

Checking that, at the initial instant ts, this Evo_Index(p) equals ZERO: Evo_Index(ts) = 0.

(%i56) Evo_Index(ts);
(%o56) 0

Derivative of Evo_Index(p) with respect to p.

(%i57) Evo_Index_derivative:'diff('Evo_Index(p),p)=diff(Evo_Index(p),p);

$$(\%o57)\ \frac{d}{d\,p}\,\mathrm{Evo\_Index}(p)=\frac{\left(\frac{d}{d\,p}\,\mathrm{M}(p)\right)\%e^{ts\,sL^2-2\,(\mathrm{M}(p)-\mathrm{M}(ts))}}{2\,\pi\,\log(2)\,Ns^2}+\frac{sL^2}{2\,\log(2)}+\frac{\frac{d}{d\,p}\,\mathrm{M}(p)}{\log(2)}$$

Thus, in general, the derivative of Evo_Index(p) is NOT A CONSTANT,
namely, in general, the Evo_Index(p) does NOT GROW LINEARLY IN TIME.

However, the last equation also reveals that:
the derivative of Evo_Index(p) is a CONSTANT (namely the growth is LINEAR)
if, and only if, the derivative of M(p) with respect to p equals ZERO.
In that case, the constant derivative equals the constant sL^2/(2*log(2)).

---

---

Checking the tsGBM case, that is the GBM STARTING AT t = ts special case of all the above formulae.
WARNING: this is MORE GENERAL than the GBM, since the GBM always starts at ts = 0.

(%i58) tsGBM_case:[M(t):=log(N0)+(muL-sL^2/2)*(t-ts), Ns=N0];

$$(\%o58)\ [\mathrm{M}(t):=\log(N0)+\left(muL-\frac{sL^2}{2}\right)(t-ts), Ns=N0]$$

The derivative of this M(t) with respect to t is:

(%i59) derivative_of_M_for_tsGBM:'diff('M(t),t)=diff(M(t),t);

$$(\%o59)\ \frac{d}{d\,t}\,\mathrm{M}(t)=muL-\frac{sL^2}{2}$$

However, in Chapter 30 of his book "Mathematical SETI", this author already proved that the (Shannon) Entropy of the GBMs decreases LINEARLY in time.
Since we know the tsGBM_Evo_Index to grow LINEARLY in time, the last derivative must equal ZERO, yielding the tsGBM CONDITION

(%i60) tsGBM_CONDITION:second(solve(rhs(derivative_of_M_for_tsGBM)=0, sL));

(%o60) $sL=\sqrt{2}\,\sqrt{muL}$

tsGBM mean value curve of t.

(%i61) def_tsGBM_mean_value:def_mean_value,tsGBM_CONDITION;

(%o61) $mean\_value=\%e^{t\ muL-ts\ muL}\ N0$

(%i62) tsGBM_mean_value:radcan(def_tsGBM_mean_value);

(%o62) $mean\_value=\%e^{(t-ts)\ muL}\ N0$

tsGBM standard deviation

(%i63) def_tsGBM_standard_deviation:def_st_dev,tsGBM_CONDITION;

(%o63) $standard\_devation=\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}\ \%e^{\frac{2\log(N0)-4\ ts\ muL+2\ t\ muL}{2}}$

(%i64) tsGBM_standard_deviation:rootscontract(factor(radcan(def_tsGBM_standard_deviation)));

(%o64) $standard\_devation=\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}\ \%e^{t\ muL-2\ ts\ muL}\ N0$

(%i65) final_GBM_upper_st_dev_curve:subst(0, ts, tsGBM_standard_deviation);

(%o65) $standard\_devation=\%e^{t\ muL}\sqrt{\%e^{2\ t\ muL}-1}\ N0$

tsGBM upper standard deviation curve of t

(%i66) def_tsGBM_upper_standard_deviation_curve:def_upper_std_curve,tsGBM_CONDITION;

(%o66) $\mathrm{upper\_std\_curve}(p)=\%e^{t\ muL-ts\ muL}\ N0+\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}\ \%e^{t\ muL-2\ ts\ muL}\ N0$

(%i67) final_tsGBM_upper_st_dev_curve:rootscontract(factor(radcan(def_tsGBM_upper_standard_deviation_cur

(%o67) $\mathrm{upper\_std\_curve}(p)=\left(\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}+\%e^{ts\ muL}\right)\%e^{t\ muL-2\ ts\ muL}\ N0$

(%i68) final_GBM_upper_st_dev_curve:subst(0, ts, final_tsGBM_upper_st_dev_curve);

(%o68) $\mathrm{upper\_std\_curve}(p)=\%e^{t\ muL}\left(\sqrt{\%e^{2\ t\ muL}-1}+1\right)N0$

tsGBM lower standard deviation curve of t

(%i69) tsGBM_lower_st_dev_curve:def_lower_std_curve,tsGBM_CONDITION;

(%o69) $\mathrm{lower\_std\_curve}(p)=\%e^{t\ muL-ts\ muL}\ N0-\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}\ \%e^{t\ muL-2\ ts\ muL}\ N0$

(%i70) final_tsGBM_lower_st_dev_curve:rootscontract(factor(radcan(tsGBM_lower_st_dev_curve)));

(%o70) $\mathrm{lower\_std\_curve}(p)=-\left(\sqrt{\%e^{2\ t\ muL}-\%e^{2\ ts\ muL}}-\%e^{ts\ muL}\right)\%e^{t\ muL-2\ ts\ muL}\ N0$

(%i71) derived_GBM_lower_st_dev_curve:subst(0, ts, final_tsGBM_lower_st_dev_curve);

(%o71) $\mathrm{lower\_std\_curve}(p)=-\%e^{t\ muL}\left(\sqrt{\%e^{2\ t\ muL}-1}-1\right)N0$

Finding the sL for tsGBM.

(%i72) def_sL;

(%o72) $sL=\dfrac{\sqrt{\log\left(\%e^{2\,\mathrm{M}(ts)}+\dfrac{DeltaNe^2\ Ns^2}{Ne^2}\right)-2\,\mathrm{M}(ts)}}{\sqrt{te-ts}}$

(%i73) def_sigma_tsGBM:def_sL,tsGBM_case;

(%o73) $sL=\dfrac{\sqrt{\log\left(\dfrac{DeltaNe^2\ N0^2}{Ne^2}+N0^2\right)-2\log(N0)}}{\sqrt{te-ts}}$

(%i74) sigma_for_tsGBM:logcontract(def_sigma_tsGBM);

(%o74) $sL=\dfrac{\sqrt{\log\left(\dfrac{Ne^2+DeltaNe^2}{Ne^2}\right)}}{\sqrt{te-ts}}$

Evo_Index(p) for the tsGBM.

(%i75) def_Evo_Index_for_tsGBM:Evo_Index(p),tsGBM_CONDITION;

(%o75) $\dfrac{(p-ts)\ muL}{\log(2)}$

(%i76) Evo_Index_for_tsGBM(p):=((p-ts)*muL)/log(2);

(%o76) $\mathrm{Evo\_Index\_for\_tsGBM}(p):=\dfrac{(p-ts)\ muL}{\log(2)}$

(%i77) derivative_of_tsGBM:'diff('Evo_Index_for_tsGBM(p),p)=diff(Evo_Index_for_tsGBM(p),p);

(%o77) $\dfrac{d}{d\,p}\,\mathrm{Evo\_Index\_for\_tsGBM}(p)=\dfrac{muL}{\log(2)}$

Re-writing the same results in the old, traditional notation muL=B and sL=sigma, Ns=A, plus ts=0 for the GBM_Evo_Index and the GBM_Peak_Locus_Theorem:

(%i78) A_exp_B_notation_for_GBM:[muL=B,sL=sigma,Ns=A];

(%o78) $[muL=B, sL=\sigma, Ns=A]$

(%i79) def_GBM_Evo_Index:'GBM_Evo_Index=Evo_Index_for_tsGBM(p),A_exp_B_notation_for_GBM;

(%o79) $GBM\_Evo\_Index=\frac{(p-ts)\,B}{\log(2)}$

(%i80) towards_GBM_Peak_Locus_Theorem_1:PEAK_LOCUS_THEOREM,tsGBM_CONDITION;

(%o80) $[\mu=\frac{\%e^{2\,ts\,muL}}{4\,\pi\,Ns^2}-p\ muL, \sigma=\frac{\%e^{ts\,muL}}{\sqrt{2}\,\sqrt{\pi}\,Ns}, b=p-\%e^{\mu-\sigma^2}]$

(%i81) towards_GBM_Peak_Locus_Theorem_2:subst(0,ts,towards_GBM_Peak_Locus_Theorem_1);

(%o81) $[\mu=\frac{1}{4\,\pi\,Ns^2}-p\ muL, \sigma=\frac{1}{\sqrt{2}\,\sqrt{\pi}\,Ns}, b=p-\%e^{\mu-\sigma^2}]$

(%i82) GBM_Peak_Locus_Theorem:towards_GBM_Peak_Locus_Theorem_2,A_exp_B_notation_for_GBM;

(%o82) $[\mu=\frac{1}{4\,\pi\,A^2}-p\ B, \sigma=\frac{1}{\sqrt{2}\,\sqrt{\pi}\,A}, b=p-\%e^{\mu-\sigma^2}]$